# 8　線形代数

*Mathematica* を用いたベクトルや行列の計算の仕方を解説する.

## ■ ベクトルと行列，次元

### ■ ベクトルと行列

ベクトルはリストで表される.

```
In[1]:= {1, 2, 3}

Out[1]= {1, 2, 3}
```

行列は各行を表すリストをリストにまとめた，２次元の配列（長さのそろったレベルの深さが２のリスト）で表される.

```
In[2]:= {{1, 2, 3}, {4, 5, 6}}

Out[2]= {{1, 2, 3}, {4, 5, 6}}
```

これは 2 行 3 列の行列である.

### ■ 行列であることのチェック　MatrixQ

行列は長さのそろったレベルの深さが２のリストでなければならい．このことを判断する関数が MatrixQ である.

```
In[3]:= MatrixQ[{{1, 2, 3}, {4, 5, 6}}]

Out[3]= True

In[4]:= MatrixQ[{{1, 2, 3}, {4, 5}}]

Out[4]= False
```

### ■ 綺麗な表示　MatrixForm

行列やベクトルを綺麗な形で表示するためには， MatrixForm を用いる.

```
In[5]:= MatrixForm[{{1, 2, 3}, {4, 5, 6}}]
```

Out[5]//MatrixForm=

$$\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{pmatrix}$$

```
In[6]:= MatrixForm[{1, 2, 3}]
```

Out[6]//MatrixForm=

$$\begin{pmatrix} 1 \\ 2 \\ 3 \end{pmatrix}$$

この例のように，ベクトルを MatrixForm で表示すると，縦ベクトル（列ベクトル）の形に表示される．横ベクトル（行ベクトル）の形に表示したければ，ベクトルをさらに {} でくくり， 1 行からなる行列にして表示すればよい．

```
In[7]:= MatrixForm[{{1, 2, 3}}]

Out[7]//MatrixForm=
        ( 1  2  3 )
```

■ 次元

行列の次元（行の数と列の数）は Dimensions で求められる．

Dimensions[mat] は行列 mat の次元を返す．

```
In[8]:= Dimensions[{{1, 2, 3}, {4, 5, 6}}]

Out[8]= {2, 3}
```

## ■ 行列を作る

行列はリストのリストなので，リストの章で見た通り Array， Table などを用いて作ることができる．その他に特別な行列を作る関数として DiagonalMatrix， IdentityMatrix がある．

### ■ Array

```
In[9]:= Array[a, {2, 3}]

Out[9]= {{a[1, 1], a[1, 2], a[1, 3]}, {a[2, 1], a[2, 2], a[2, 3]}}
```

これは第 i， j 成分が a[i， j] であるような 2×3 次の行列である．

### ■ Table

次は第 i， j 成分が i+j であるような 2×3 次の行列である．

```
In[10]:= Table[i + j, {i, 2}, {j, 3}]

Out[10]= {{2, 3, 4}, {3, 4, 5}}

In[11]:= MatrixForm[%]

Out[11]//MatrixForm=
        ( 2  3  4 )
        ( 3  4  5 )
```

次は乱数を成分にもつ行列である．

```
In[12]:= Table[Random[], {2}, {3}]

Out[12]= {{0.803894, 0.687776, 0.0849148}, {0.945497, 0.677564, 0.740034}}
```

特に零行列（すべての成分が 0 であるような行列）は，次で得られる．

```
In[13]:= Table[0, {2}, {3}]

Out[13]= {{0, 0, 0}, {0, 0, 0}}
```

練習問題： 第 $i,\ j$ 成分が $a_{ij} = i - j$ であるような $3 \times 3$ の行列を作れ.

## ■ DiagonalMatrix

対角成分以外は 0 であるような正方行列は DiagonalMatrix で作れる.

DiagonalMatrix[{a1, a2, ...}] は a1, a2, ... を対角成分とする対角行列を返す.

```
In[14]:= DiagonalMatrix[{a, b, c}]

Out[14]= {{a, 0, 0}, {0, b, 0}, {0, 0, c}}

In[15]:= MatrixForm[%]

Out[15]//MatrixForm=
```

$$\begin{pmatrix} a & 0 & 0 \\ 0 & b & 0 \\ 0 & 0 & c \end{pmatrix}$$

## ■ IdentityMatrix

単位行列は IdentityMatrix で作れる.

IdentityMatrix[n] は n×n の単位行列を返す.

```
In[16]:= IdentityMatrix[3]

Out[16]= {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}

In[17]:= MatrixForm[%]

Out[17]//MatrixForm=
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

## ■ 転置行列をとる

行列 $A$ の転置行列 ${}^tA$ （行と列を入れ替えた行列）を求めるには， Transpose を用いる.

Transpose[mat] は行列 mat の転置行列を返す.

```
In[1]:= MatrixForm[mat = Array[a, {3, 4}]]

Out[1]//MatrixForm=
```

$$\begin{pmatrix} a[1,1] & a[1,2] & a[1,3] & a[1,4] \\ a[2,1] & a[2,2] & a[2,3] & a[2,4] \\ a[3,1] & a[3,2] & a[3,3] & a[3,4] \end{pmatrix}$$

```
In[19]:= MatrixForm[Transpose[mat]]

Out[19]//MatrixForm=
```

$$\begin{pmatrix} a[1,1] & a[2,1] & a[3,1] \\ a[1,2] & a[2,2] & a[3,2] \\ a[1,3] & a[2,3] & a[3,3] \\ a[1,4] & a[2,4] & a[3,4] \end{pmatrix}$$

## ■ 行列の部分を取り出す

行列の部分を取り出すには，リストのときと同じように Part などを用いればよいが，行列の場合に特に有効な方法があるので，次の行列 mat を例にとりそれを説明する．

```
In[24]:= MatrixForm[mat]

Out[24]//MatrixForm=
```

$$\begin{pmatrix} a[1,1] & a[1,2] & a[1,3] & a[1,4] \\ a[2,1] & a[2,2] & a[2,3] & a[2,4] \\ a[3,1] & a[3,2] & a[3,3] & a[3,4] \end{pmatrix}$$

## ■ 成分を取り出す

Part[mat, i, j] または mat[[i, j]] は，行列 mat の第 i, j 成分を返す．

```
In[25]:= mat[[1, 2]]

Out[25]= a[1, 2]
```

Part[mat, {i1, i2, ...}, {j1, j2, ...}] または mat[[{i1, i2, ...}, {j1, j2, ...}]] は，第 i1, i2, ... 行と第 j1, j2, ... 列の場所から取り出した mat の小行列を返す．

```
In[26]:= Part[mat, {1, 3}, {1, 2, 4}]

Out[26]= {{a[1, 1], a[1, 2], a[1, 4]}, {a[3, 1], a[3, 2], a[3, 4]}}

In[27]:= MatrixForm[%]

Out[27]//MatrixForm=
```

$$\begin{pmatrix} a[1,1] & a[1,2] & a[1,4] \\ a[3,1] & a[3,2] & a[3,4] \end{pmatrix}$$

## ■ 行を取り出す

行を取り出すのは簡単である．

Part[mat, i] または mat[[i]] は，行列 mat の第 i 行を返す．

```
In[28]:= mat[[2]]

Out[28]= {a[2, 1], a[2, 2], a[2, 3], a[2, 4]}
```

## ■ 列を取り出す

列を取り出すには，一度転置行列をとってからその行を取り出せばよい．

Transpose[mat][[j]] は行列 mat の第 j 列を取り出す.

```
In[29]:= Transpose[mat][[2]]

Out[29]= {a[1, 2], a[2, 2], a[3, 2]}
```

## ■ ベクトルや行列の和差とスカラー倍

ベクトルや行列の和差やスカラー倍を求めるには，通常通り +, - や * （またはスペース" "）を用いればよい.

### ■ ベクトルや行列の和差

```
In[30]:= {1, 2, 3} + {4, 5, 6}

Out[30]= {5, 7, 9}

In[31]:= {5, 5, 5 } - {1, 2, 3}

Out[31]= {4, 3, 2}

In[32]:= {{1, 2}, {3, 4}} + {{5, 6}, {7, 8}}

Out[32]= {{6, 8}, {10, 12}}
```

### ■ ベクトルや行列のスカラー倍

```
In[33]:= 3 * {1, 2, 3}

Out[33]= {3, 6, 9}

In[34]:= 3 {{1, 2}, {3, 4}}

Out[34]= {{3, 6}, {9, 12}}
```

### ■ 注意事項

{1, 2, 3} + 4 などという演算は数学的には意味がないが，*Mathematica* は {1, 2, 3} のすべての要素に 4 を加えた {1+4, 2+4, 3+4} を返す．これは思わぬ計算間違いの原因となるので，注意が必要である.

```
In[35]:= {1, 2, 3} + 4

Out[35]= {5, 6, 7}
```

## ■ 行列の積，行列とベクトルの積，内積

行列の積，行列とベクトルの積，ベクトルの内積はすべて，ドット演算子あるいはドット積と呼ばれる記号で行う．それはピリオド "." であるが，関数の名前は Dot である.

## ■ 行列と行列の積

Dot[mat1, mat2] あるいは mat1.mat2 は，二つの行列 mat1 と mat2 の積を返す．

$l \times m$ 次の行列 $(a_{i\,j})$と $m \times n$ 次の行列 $(b_{i\,j})$ との積は $l \times n$ の行列 $(c_{i\,j})$ であって，その定義は

$$c_{i\,j} = \sum_{k=1}^{m} a_{i\,k}\, b_{k\,j}$$

である．

次は $\begin{pmatrix} 1 & 2 \\ 2 & 2 \end{pmatrix} \begin{pmatrix} 1 & 1 & 0 \\ 2 & 1 & 1 \end{pmatrix} = \begin{pmatrix} 5 & 3 & 2 \\ 6 & 4 & 2 \end{pmatrix}$ の計算である．

```
In[36]:= {{1, 2}, {2, 2}} . {{1, 1, 0}, {2, 1, 1}}

Out[36]= {{5, 3, 2}, {6, 4, 2}}
```

次の計算 $\begin{pmatrix} 1 & 2 & 3 \\ 2 & 3 & 4 \end{pmatrix} \begin{pmatrix} 1 & 2 \\ 3 & 4 \end{pmatrix}$ は，左側の行列の列の個数と右側の行列の行の個数とが異なるので，積が定義できない．

```
In[37]:= {{1, 2, 3}, {2, 3, 4}} . {{1, 2}, {3, 4}}

        Dot::dotsh : Tensors {{1, 2, 3}, {2, 3, 4}} and {{1, 2}, {3, 4}} have incompatible shapes.

Out[37]= {{1, 2, 3}, {2, 3, 4}}.{{1, 2}, {3, 4}}
```

## ■ 行列とベクトルの積，ベクトルと行列の積

Dot[mat, vec] あるいは mat.vec は，行列 mat とベクトル vec の積を返す．

$m \times n$ 次の行列 $(a_{i\,j})$と $n$ 次の縦ベクトル $(b_i)$ との積は $m$ のベクトル $(c_i)$ であって，その定義は

$$c_i = \sum_{k=1}^{n} a_{i\,k}\, b_k$$

である．

次は $\begin{pmatrix} 1 & 2 & 0 \\ 2 & 0 & 1 \end{pmatrix} \begin{pmatrix} 1 \\ 1 \\ 2 \end{pmatrix} = \begin{pmatrix} 3 \\ 4 \end{pmatrix}$ の計算である．

```
In[38]:= {{1, 2, 0}, {2, 0, 1}} . {1, 1, 2}

Out[38]= {3, 4}
```

Dot[vec, mat] あるいは vec.mat は，ベクトル vec と行列 mat の積を返す．

$m$ 次の横ベクトル $(b_i)$ と $m \times n$ 次の行列 $(a_{i\,j})$との積は $n$ のベクトル $(c_i)$ であって，その定義は

$$c_i = \sum_{k=1}^{m} b_k\, a_{k\,i}$$

である．

次は $\begin{pmatrix} 2 & 1 \end{pmatrix} \begin{pmatrix} 1 & 2 & 0 \\ 2 & 0 & 1 \end{pmatrix} = \begin{pmatrix} 4 & 4 & 1 \end{pmatrix}$

```
In[39]:= {2, 1}.{{1, 2, 0}, {2, 0, 1}}

Out[39]= {4, 4, 1}
```

ここで注意すべきことがある．それは *Mathematica* では縦ベクトルも横ベクトルも同じ一重のリストで表されていることである．すなわち，縦ベクトルと横ベクトルを区別しないわけである．それでいて，ベクトルがドット演算子の右にあるときには縦ベクトル，左にあるときには横ベクトルとして，ちゃんと計算できている理由は次の通りである．

$a_{i_1,i_2,\ldots,i_k}$ をインデックス $i_1$，$i_2$，...，$i_k$ の要素とする $m_1 \times m_2 \times \cdots \times m_k$ 型の配列 a と $b_{j_1,j_2,\ldots,j_l}$ をインデックス $j_1$，$j_2$，...，$j_l$ の要素とする $n_1 \times n_2 \times \cdots \times n_l$ 型の配列 b とに対し，そのドット積 a.b は， a の一番内側のインデックス $i_k$ と b の一番外側のインデックス $j_1$ とに関して，掛けて和をとるという縮約を行い，$m_1 \times m_2 \times \cdots \times m_{k-1} \times n_2 \times \cdots \times n_l$ 型の配列を返す．

したがって，{{1, 2, 0}, {2, 0, 1}} . {1, 1, 2} という 2×3 型の配列と 3 型の配列とのドット積は 2 型の配列すなわち 2 次元のベクトルを返し， {2, 1}.{{1, 2, 0}, {2, 0, 1}} という 2 型の配列と 2×3 型の配列のドット積は 3 型の配列すなわち 3 次元のベクトルを返す．

## ■ ベクトルの内積

`Dot[vec1, vec2]` あるいは `vec1.vec2` は，二つのベクトル `vec1` と `vec2` の内積を返す．

二つの $m$ 次のベクトル $(a_i)$ と $(b_i)$との内積はスカラー（数値）であって，その定義は

$$\sum_{i=1}^{m} a_i\, b_i$$

である．

```
In[40]:= {1, 2, 3} . {2, 3 , 4}

Out[40]= 20
```

## ■ ベクトルの大きさ

ベクトル `v` の大きさ（長さ）は `Sqrt[v.v]` で求められる．

```
In[41]:= Sqrt[{1, 2, 3}.{1, 2, 3}]
```

Out[41]= $\sqrt{14}$

## ■ 注意事項

ドット積 "." ではなく，通常の数の積 "*" （あるは空白 " "）を用いて行列やベクトルの積をとると，単に同じ位置にある成分を掛けてできるものとなる．

```
In[39]:= {a, b, c} * {x, y, z}

Out[39]= {a x, b y, c z}
```

練習問題： `matA, matB, matC` を，第 `i, j` 成分が各々 `a[i, j], b[i, j], c[i, j]` である 2 × 2 の行列とせよ． `(matA . matB) . matC` と `matA . (matB . matC)` とを各々計算せよ．その二つの結果が等しいことを，引き算して零行列になることで確かめよ．

## ■ 正方行列のベキ乗

正方行列 $A$ に対して，そのベキ乗 $A^n$ は `MatrixPower` を用いて計算できる．

`MatrixPower[mat, n]` は行列 `mat` のドット演算子による `n` 乗，すなわち `mat.mat.⋯.mat` を返す．

```
In[45]:= MatrixForm[mat = {{1, 1}, {2, 1}}]
```

Out[45]//MatrixForm=

$$\begin{pmatrix} 1 & 1 \\ 2 & 1 \end{pmatrix}$$

`mat` に `{{1, 1}, {2, 1}}` を割り当てて，その値を `MatrixForm` で表示している.

次は `mat` の 3 乗である.

```
In[46]:= MatrixForm[MatrixPower[mat, 3]]
```

Out[46]//MatrixForm=

$$\begin{pmatrix} 7 & 5 \\ 10 & 7 \end{pmatrix}$$

`MatrixPower` は，固有ベクトルを用いて行列を対角化しベキ乗を求めるので， `n` 乗の一般式を得ることができる.

```
In[48]:= MatrixForm[MatrixPower[mat, n]]
```

Out[48]//MatrixForm=

$$\begin{pmatrix} \frac{1}{2}(1-\sqrt{2})^n + \frac{1}{2}(1+\sqrt{2})^n & -\frac{(1-\sqrt{2})^n}{2\sqrt{2}} + \frac{(1+\sqrt{2})^n}{2\sqrt{2}} \\ -\frac{(1-\sqrt{2})^n}{\sqrt{2}} + \frac{(1+\sqrt{2})^n}{\sqrt{2}} & \frac{1}{2}(1-\sqrt{2})^n + \frac{1}{2}(1+\sqrt{2})^n \end{pmatrix}$$

## ■ 注意事項

行列 `mat` のベキ乗を求めるのに `mat^n` とすると，誤った答えとなるので注意するように.

```
In[40]:= {{a, b}, {c, d}}^n
```

Out[40]= $\{\{a^n, b^n\}, \{c^n, d^n\}\}$

練習問題：　行列 $\begin{pmatrix} 1 & 1 \\ 2 & 0 \end{pmatrix}$ の `n` 乗を `MatrixPower` を用いて求めよ.

## ■ 正方行列の逆行列

正方行列 $A$ に対して，その逆行列 $A^{-1}$ は `Inverse` を用いて計算できる.

`Inverse[mat]` は正方行列 `mat` の逆行列を返す.

```
In[4]:= Inverse[{{1, 2}, {3, 4}}]
```

Out[4]= $\left\{\{-2, 1\}, \left\{\frac{3}{2}, -\frac{1}{2}\right\}\right\}$

逆行列をかけると単位行列になる.

```
In[5]:= {{1, 2}, {3, 4}}.%
```

Out[5]= {{1, 0}, {0, 1}}

行列はいつでも逆行列をもつとは限らない.

```
In[54]:= Inverse[{{1, 2}, {2, 4}}]

        Inverse::sing : Matrix {{1, 2}, {2, 4}} is singular.

Out[54]= Inverse[{{1, 2}, {2, 4}}]
```

練習問題： $\begin{pmatrix} 1 & 1 & 2 \\ 2 & 1 & 2 \\ 1 & 0 & 2 \end{pmatrix}$ の逆行列を求めよ．また，それを $\begin{pmatrix} 1 & 1 & 2 \\ 2 & 1 & 2 \\ 1 & 0 & 2 \end{pmatrix}$ に掛けて単位行列になることを確認せよ．

## ■ 正方行列の行列式

正方行列 $A$ に対して，その行列式 $\det(A)$ は Det を用いて計算できる．

### ■ Det

Det[mat] は正方行列 mat の行列式の値を返す．

```
In[55]:= Det[{{a, b}, {c, d}}]

Out[55]= -b c + a d

In[56]:= Det[{{a, b, c}, {d, e, f}, {g, h, i}}]

Out[56]= -c e g + b f g + c d h - a f h - b d i + a e i
```

先ほどの逆行列をもたない行列の行列式は 0 である．

```
In[57]:= Det[{{1, 2}, {2, 4}}]

Out[57]= 0
```

練習問題： matA, matB を，第 i, j 成分が各々 a[i, j], b[i, j] である 2 × 2 の行列とせよ． matA . matB の行列式を Factor を用いて因数分解して，それが matA の行列式と matB の行列式の積であることを確認せよ．

## ■ 正方行列の固有値と固有ベクトル

正方行列 $A$ と $0$ ベクトルでないベクトル $x$ に対して， $A x = \lambda x$ であるとき $\lambda$ を $A$ の固有値， $x$ を $A$ の固有ベクトルという．これらは Eigenvalues, Eigenvectors または Eigensystem を用いて得られる．以下，次の行列 mat を例にして解説する．

```
In[58]:= MatrixForm[mat]

Out[58]//MatrixForm=
```

$$\begin{pmatrix} 1 & 1 \\ 2 & 1 \end{pmatrix}$$

## ■ Eigenvalues

`Eigenvalues[mat]` は正方行列 `mat` の固有値のリストを返す.

In[59]:= **eval = Eigenvalues[mat]**

Out[59]= $\{1-\sqrt{2}, 1+\sqrt{2}\}$

## ■ Eigenvectors

`Eigenvectors[mat]` は正方行列の固有ベクトルのリストを返す.

In[60]:= **evec = Eigenvectors[mat]**

Out[60]= $\left\{\left\{-\frac{1}{\sqrt{2}}, 1\right\}, \left\{\frac{1}{\sqrt{2}}, 1\right\}\right\}$

行列 `mat` と 1 番目の固有ベクトルとのドット積をとってみる.

In[61]:= **mat.evec[[1]]**

Out[61]= $\left\{1-\frac{1}{\sqrt{2}}, 1-\sqrt{2}\right\}$

これは 1 番目の固有ベクトルの 1 番目の固有値倍に等しい.

In[62]:= **evec[[1]] eval[[1]]**

Out[62]= $\left\{-\frac{1-\sqrt{2}}{\sqrt{2}}, 1-\sqrt{2}\right\}$

一見異なっているが, `Simplify` を用いて簡約化すれば一致している.

In[63]:= **Simplify[%]**

Out[63]= $\left\{1-\frac{1}{\sqrt{2}}, 1-\sqrt{2}\right\}$

## ■ Eigensystem

`Eigensystem[mat]` は正方行列の固有値と固有ベクトルとを同時に求め, そのリストを返す.

In[64]:= **Eigensystem[mat]**

Out[64]= $\left\{\{1-\sqrt{2}, 1+\sqrt{2}\}, \left\{\left\{-\frac{1}{\sqrt{2}}, 1\right\}, \left\{\frac{1}{\sqrt{2}}, 1\right\}\right\}\right\}$

## ■ 線形方程式

線形方程式（連立一次方程式）は係数行列 $A$ を用いて, $Ax=b$ という行列とベクトル関する方程式になる. このような方程式は `LinearSolve`, `NullSpace` を用いて解くことができる.

## ■ LinearSolve

LinearSolve[m, b] は，行列方程式 m.x == b を満たすベクトル x の一つ（特殊解）を返す．

```
In[65]:= MatrixForm[m = {{1, 2, 3, 4}, {1, 1, 1, 1}}]
```

Out[65]//MatrixForm=

$$\begin{pmatrix} 1 & 2 & 3 & 4 \\ 1 & 1 & 1 & 1 \end{pmatrix}$$

```
In[66]:= MatrixForm[b = {2, 3}]
```

Out[66]//MatrixForm=

$$\begin{pmatrix} 2 \\ 3 \end{pmatrix}$$

次の x0 は $\begin{pmatrix} 1 & 2 & 3 & 4 \\ 1 & 1 & 1 & 1 \end{pmatrix} x = \begin{pmatrix} 2 \\ 3 \end{pmatrix}$ の解の一つである．

```
In[67]:= MatrixForm[x0 = LinearSolve[m, b]]
```

Out[67]//MatrixForm=

$$\begin{pmatrix} 4 \\ -1 \\ 0 \\ 0 \end{pmatrix}$$

確かめてみると，確かに m.x0 == b となっている．

```
In[68]:= MatrixForm[m.x0]
```

Out[68]//MatrixForm=

$$\begin{pmatrix} 2 \\ 3 \end{pmatrix}$$

## ■ NullSpace

NullSpace[m] は， m の零空間（行列方程式 m.x == 0 の解空間）の一組の基底を返す．

先ほどの行列 m については， m の零空間は次の二つのベクトルを基底にもつ．

```
In[69]:= ns = NullSpace[m]

Out[69]= {{2, -3, 0, 1}, {1, -2, 1, 0}}
```

この二つのベクトルは m.x == 0 の解である．

```
In[70]:= m.ns[[1]]

Out[70]= {0, 0}

In[71]:= m.ns[[2]]

Out[71]= {0, 0}
```

## ■ 行列方程式の一般解

行列方程式 m.x == b の一般解 x は，LinearSolve で求めた特殊解 x0 と NullSpace で求めた m の零空間の基底 ns と，基底のベクトルの個数だけのパラメータ s, t とで，次のように求めることができる．

```
In[72]:= MatrixForm[x = x0 + s ns[[1]] + t ns[[2]]]
```

```
Out[72]//MatrixForm=
```

$$\begin{pmatrix} 4+2s+t \\ -1-3s-2t \\ t \\ s \end{pmatrix}$$

s ns[[1]] + t ns[[2]] は {s, t}.ns としてもよい．

これが m.x == b すなわち m.x - b == 0 の解になっていることを確かめる．

```
In[73]:= Simplify[m.x - b]

Out[73]= {0, 0}
```

## ■ 演習問題

[8-1] 第 $i$, $j$ 成分が $i^2 - i\,j + 3\,j^2$ であるような 3×3 の行列 mat を作れ．また，その行列式と逆行列を求めよ．

[8-2] [8-1] で作った mat に対して，その転置行列を求めよ．また，その行列式と逆行列を求めよ．

[8-3] 二つの 3×3 次の正方行列 $A$, $B$ に対して，${}^t(A\,B) = {}^tB\,{}^tA$ であることを，*Mathematica* を用いて証明せよ．

[8-4] 二つのベクトルの成す角 $\theta$ の余弦は $\cos\theta = \frac{u\cdot v}{|u|\,|v|}$ で求められる．ここで，$u\cdot v$ は内積，$|u|$ はベクトル $u$ の大きさである．(1, 3, 1), (-2, 1, 3) という二つのベクトルの成す角の余弦を求めよ．

[8-5] 行列方程式 $\begin{pmatrix} 3 & 2 & 1 & 1 \\ 2 & 1 & 2 & 3 \end{pmatrix}\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} 1 \\ 2 \end{pmatrix}$ の一般解を，パラメータ s, t を用いて表せ．